

RCSL: 43-RI0314

AUTHOR: LISC

EDITED: 76.04.28

APPLICATION PROCEDURE RC36-R0013.01

MUSIL CODEPROCEDURE ACCES

KEYWORDS: MUSIL, APPLICATION, CODEPROCEDURE, DATA ENTRY, LISTING

ABSTRACT: THE PROCEDURE READS A PART OF A RECORD FROM ZONE Z.  
BEFORE THE FIRST CALL OF A BATCH THE ZONE MUST BE  
POSITIONED TO THE FIRST BYTE IN THE FIRST DATA BLOCK.

RCSL: 43-RI0315: ASCII SOURCE TAPE.

RCSL: 43-RI0316: REL. BIN. TAPE.

; LISC 28.04.76 CODEPROCEDURE ACCES

...01...

.TITL ACCES  
 .TXTM 1  
 .RDX 10  
 .NREL

VERSION=01 ; 28.04.76

```
; CODEPROCEDURE ACCES(FILE Z;
;           VAR FIELD   :STRING(80);
;           VAR NO      :INTEGER;
;           VAR STATUS  :INTEGER;
;           VAR SUBFN   :STRING(2));
```

```
; THE PROCEDURE READS A PART OF A RECORD FROM ZONE Z.
; BEFORE THE FIRST CALL ON A BATCH THE ZONE MUST BE POSITIONED TO
; THE FIRST BYTE IN THE FIRST DATA BLOCK.
; THE RECORD PART ACCESSED IS DESCRIBED IN NO:
;   NO>=0:  A FIELD,
;           - THE FIELD CONTENTS IS PLACED AS A STRING IN FIELD
;           - STATUS HOLDS THE FIELD STATUS
;           - SUBFN IS UNDEFINED
;   NO=-1:  A RECORD HEAD,
;           - THE RECORD STATUS IS PLACED IN STATUS
;           - THE SUBFORMAT NAME IS PLACED AS A STRING IN SUBFN
;           - FIELD IS UNDEFINED
;   NO=-2:  A RECORD END,
;           - FIELD, STATUS AND SUBFN ARE UNDEFINED
;   NO=-3:  A BATCH END,
;           - FIELD, STATUS AND SUBFN ARE UNDEFINED
```

```
; NOTE:
; THE ZONE MUST NOT BE CHANGED BY THE CALLER BETWEEN FIRST AND
; LAST CALL OF ACCES.
```

```
FDBLK= 1           ; FIRST DATA BLOCK
```

```

ACCES:                                ;ACCES:
JSR@  MZSTART+2 ;
STA   1,RPZON  ; Z:= FIRST PARAM, ADDRESS;
JSR@  MZSTART+2 ;
STA   1,RPFLD  ; FIELD:= SECOND PARAM, BYTEADDRESS;
JSR@  MZSTART+2 ;
STA   1,RPNO   ; NO:= THIRD PARAM, ADDRESS;
JSR@  MZSTART+2 ;
STA   1,RPSTA  ; STATUS:= FOURTH PARAM, ADDRESS;
JSR@  MZSTART+2 ;
STA   1,RPSUB  ; SUBFN:= FIFTH PARAM, BYTEADDRESS;
LDA   2,RPZON  ;
LDA   0,ZBLOCK,2;
LDA   1,ZREM,2 ;
LDA   2,RP1BL  ;
IEQ   0,2      ; IF Z.ZBLOCK = FIRST DATA BLOCK AND
MOV#  1,1,SZR  ;     Z.ZREM = 0 THEN
JMP   RP001   ;
LDA   0,.0    ;
STA   0,RPSTT ; STATE:= AFTER RECMARK;
RP001: ;
LDA   2,RPZON ;
LDA   0,RPSTT ;
MOV#  0,0,SZR ; IF STATE = AFTER RECMARK THEN
JMP   RP004   ; BEGIN
INCHAR ;     INCHAR(Z,CHAR);
LDA   0,RPA62 ;
SEQ   0,1     ; IF CHAR = BATCH END MARK THEN
JMP   RP002   ; BEGIN
LDA   0,.M3   ;
STA@  0,RPNO  ;     NO:= -3;
LDA   0,.0    ;
STA   0,RPSTT ;     STATE:= AFTER RECMARK;
;     END
JMP   RP003   ; ELSE
RP002: ; BEGIN C. RECORD HEAD;
MOVS  1,1     ;
LDA   3,RPSUB ;
MOVZR 3,3     ;
STA   1,0,3   ;     SUBFN(1):= CHAR<<8 + 0;
INCHAR ;     INCHAR(Z,CHAR);
STA@  1,RPSTA ;     STATUS:= CHAR;
ADC   0,0     ;
STA@  0,RPNO  ;     NO:= -1;
LDA   0,.1    ;
STA   0,RPSTT ;     STATE:= AFTER FIELDMARK;
RP003: ;     END;
;     END
JMP   RP010   ; ELSE
  
```

```
RP004: ;
LDA 1,.1 ;
SEQ 0,1 ; IF STATE = AFTER FIELDMARK THEN
JMP RP009 ; BEGIN
INCHAR ; INCHAR(Z,CHAR);
STA@ 1,RPSTA ; STATUS:= CHAR;
INCHAR ; INCHAR(Z,CHAR);
LDA 0,RPA60 ;
SEQ 0,1 ; IF CHAR = RECORD END MARK THEN
JMP RP005 ; BEGIN
ADCZL 0,0 ;
STA@ 0,RPNO ; NO:= -2;
LDA 0,.0 ;
STA 0,RPSTT ; STATE:= AFTER RECMARK;
; END
JMP RP008 ; ELSE
RP005: ; BEGIN C. FIELD;
LDA 0,.0 ;
STA@ 0,RPNO ; NO:= 0;
RP006: ; WHILE
LDA 0,RPA61 ; CHAR <>
SNE 0,1 ; FIELD END MARK DO
JMP RP007 ; BEGIN
MOV 1,0 ;
LDA 1,RPFLD ;
PUTBYTE ; FIELD(NO:= CHAR;
ISZ RPFLD ;
ISZ@ RPNO ; NO:= NO + 1;
LDA 2,RPZON ;
INCHAR ; INCHAR(Z,CHAR);
JMP RP006 ; END;
RP007: ;
LDA 0,.1 ;
STA 0,RPSTT ; STATE:= AFTER FIELDMARK;
RP008: ; END;
RP009: ; END;
RP010: ;
LDA 2,CUR ;
JSR@ MZSTART+1 ; RETURN;
```

RPZON: 0 ; ADDRESS OF Z  
RPFLD: 0 ; BYTEADDRESS OF FIELD  
RPNO: 0 ; ADDRESS OF NO  
RPSTA: 0 ; ADDRESS OF STATUS  
RPSUB: 0 ; BYTEADDRESS OF STATUS  
RP1BL: FDBLK ; FIRST DATA BLOCK  
RPSTT: 0 ; STATE, INIT = AFTER RECMARK  
RPA62: AA62 ; VALUE OF BATCH END MARK  
RPA60: AA60 ; VALUE OF RECORD END MARK  
RPA61: AA61 ; VALUE OF FIELD END MARK

.END ACCES ; END CODEPROCEDURE ACCES

